
Practice with Fixed Effects Interactions: COMPLETED version 
Example Outcome = Income (in 1000s) Predicted from Sex (MW) and Years Education (ed12) 

Everything’s significant in this simulated data, so no p-values are provided.  

 
************************** Model 1 ***************************************; 
TITLE "First Model: Fixed Quadratic Time, Time1=0 and Men=0 and Ed12=0"; 
PROC MIXED DATA=example COVTEST NOCLPRINT NOITPRINT METHOD=ML; 
 CLASS ID; 
 MODEL income = t1 t1*t1 MW ed12 MW*ed12 / SOLUTION DDFM=Satterthwaite;  
 RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;  
 
* In ESTIMATE statements, first entry is the effect you are estimating  
  (so first is always 1); 
 
* Example of asking for predicted values at specific levels of predictors; 
* Here, "t1" is replaced with actual time value; 
 
ESTIMATE "Intercept at T1 for Men   Ed=12"  Int 1 t1 0 t1*t1 0  MW 0 ed12 0 MW*ed12 0; 
ESTIMATE "Intercept at T1 for Men   Ed=16"  Int 1 t1 0 t1*t1 0  MW 0 ed12 4 MW*ed12 0; 
ESTIMATE "Intercept at T1 for Women Ed=12"  Int 1 t1 0 t1*t1 0  MW 1 ed12 0 MW*ed12 0; 
ESTIMATE "Intercept at T1 for Women Ed=16"  Int 1 t1 0 t1*t1 0  MW 1 ed12 4 MW*ed12 4; 
 
ESTIMATE "Intercept at T6 for Men   Ed=12"  Int 1 t1 5 t1*t1 25 MW 0 ed12 0 MW*ed12 0; 
ESTIMATE "Intercept at T6 for Men   Ed=16"  Int 1 t1 5 t1*t1 25 MW 0 ed12 4 MW*ed12 0; 
ESTIMATE "Intercept at T6 for Women Ed=12"  Int 1 t1 5 t1*t1 25 MW 1 ed12 0 MW*ed12 0; 
ESTIMATE "Intercept at T6 for Women Ed=16"  Int 1 t1 5 t1*t1 25 MW 1 ed12 4 MW*ed12 4; 
 
 
* Example of asking for slopes at particular occasions; 
* Here one of the "t1"=1, the other t1=actual time; 
 
ESTIMATE "Linear Slope (for anyone) at T1"  t1 1 t1*t1 0; 
ESTIMATE "Linear Slope (for anyone) at T6"  t1 1 t1*t1 10; * Remember, quad*2; 
 
 
* Example of asking for simple main effects of an interaction; 
 
ESTIMATE "Sex Diff (at any time) for Ed=12"   MW 1 MW*ed12 0; 
ESTIMATE "Sex Diff (at any time) for Ed=16"   MW 1 MW*ed12 4; 
 
ESTIMATE "Ed Slope (at any time) for Men" ed12 1 MW*ed12 0; 
ESTIMATE "Ed Slope (at any time) for Women"  ed12 1 MW*ed12 1; 
 
 
run; 
 
 
* Can check estimates by re-centering predictors; 
TITLE "First Model: Time6=0 and Women=0 and Ed16=0 Instead"; 
PROC MIXED DATA=example COVTEST NOCLPRINT NOITPRINT METHOD=ML; 
 CLASS ID; 
 MODEL income = t6 t6*t6 WM ed16 WM*ed16 / SOLUTION DDFM=Satterthwaite;  
 RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;  
run; 

WHAT SAS GIVES YOU BY DEFAULT IF TIME1=0, MEN=0, AND ED12=0:  

       Solution for Fixed Effects 
                         Standard 
Effect       Estimate       Error       
 
Intercept     49.0432     0.07535  expected income of 49k at T1 for Man with Ed=12 
t1             4.1689     0.05904  at T1, income increases 4.2k/time for all  
t1*t1         -0.1929     0.01133  linear rate of change/time decreases by .4/time for all    
MW            -8.4099     0.06495  if ed=12, women make 8.4k less than men   
ed12           2.2751     0.02523  in men, income increases by 2.3k/year of ed   
MW*ed12       -0.3983     0.03269  gender gap grows by .4/year OR ed slope is .4 smaller in women   
 

WHAT EXTRA CONDITIONAL PREDICTIONS AND EFFECTS WE ASKED FOR: 

                                      Estimates 
                                               Standard 
Label                              Estimate       Error      
 
Intercept at T1 for Men   Ed=12     49.0432     0.07535     
Intercept at T1 for Men   Ed=16     58.1438      0.1241     
Intercept at T1 for Women Ed=12     40.6333     0.06944     
Intercept at T1 for Women Ed=16     48.1407      0.1092     
 
Intercept at T6 for Men   Ed=12     65.0664     0.07535     
Intercept at T6 for Men   Ed=16     74.1670      0.1241     
Intercept at T6 for Women Ed=12     56.6565     0.06944     
Intercept at T6 for Women Ed=16     64.1639      0.1092     
 
Linear Slope (for anyone) at T1      4.1689     0.05904     
Linear Slope (for anyone) at T6      2.2403     0.05904     
 
Sex Diff (at any time) if Ed=12     -8.4099     0.06495     
Sex Diff (at any time) if Ed=16    -10.0031      0.1451     
 
Ed Slope (at any time) for Men       2.2751     0.02523     
Ed Slope (at any time) for Women     1.8768     0.02078     

 
WHAT YOU GET IF YOU RE-CENTER AT TIME6=0, WOMEN=0, AND ED16=0 INSTEAD: 

       Solution for Fixed Effects 
                         Standard 
Effect       Estimate       Error       
 
Intercept     64.1639      0.1092  expected income of 64k at T6 for women with ed=16   
t6             2.2403     0.05904  at T6, income increases by 2.2k/time for all   
t6*t6         -0.1929     0.01133  linear rate of change/time decreases by .4/time for all   
WM            10.0031      0.1451  if ed=16, men make 10k more than women   
ed16           1.8768     0.02078  in women, income increases by 1.9k/year of ed   
WM*ed16        0.3983     0.03269  gender gap grows by .4/year or ed slope is .4 bigger in men  

************************** Model 2 ***************************************; 
TITLE "Second Model: Fixed Quadratic Time, Time1=0 and Men=0 and Ed12=0"; 
PROC MIXED DATA=example COVTEST NOCLPRINT NOITPRINT METHOD=ML; 
 CLASS ID; 
 MODEL income = t1 t1*t1 MW ed12 MW*ed12  
      MW*t1 MW*t1*t1 ed12*t1 / SOLUTION DDFM=Satterthwaite;  
 RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;  
 
* Example of asking for slopes at particular occasions for particular people; 
* Here one of the "t1" = 1, the other t1=actual time; 
 
ESTIMATE "Linear at T1 for Men   Ed=12" t1 1 t1*t1 0  MW*t1 0 MW*t1*t1 0  ed12*t1 0; 
ESTIMATE "Linear at T1 for Men   Ed=16" t1 1 t1*t1 0  MW*t1 0 MW*t1*t1 0  ed12*t1 4; 
ESTIMATE "Linear at T1 for Women Ed=12" t1 1 t1*t1 0  MW*t1 1 MW*t1*t1 0  ed12*t1 0; 
ESTIMATE "Linear at T1 for Women Ed=16" t1 1 t1*t1 0  MW*t1 1 MW*t1*t1 0  ed12*t1 4; 
 
ESTIMATE "Linear at T6 for Men   Ed=12" t1 1 t1*t1 10 MW*t1 0 MW*t1*t1 0  ed12*t1 0; 
ESTIMATE "Linear at T6 for Men   Ed=16" t1 1 t1*t1 10 MW*t1 0 MW*t1*t1 0  ed12*t1 4; 
ESTIMATE "Linear at T6 for Women Ed=12" t1 1 t1*t1 10 MW*t1 1 MW*t1*t1 10 ed12*t1 0; 
ESTIMATE "Linear at T6 for Women Ed=16" t1 1 t1*t1 10 MW*t1 1 MW*t1*t1 10 ed12*t1 4; 
 
 
* Example of asking for simple main effects of an interaction; 
* Here, "t1" is actual time always; 
 
ESTIMATE "Sex Diff at T1 if Ed=12" MW 1 MW*t1 0 MW*t1*t1 0  MW*ed12 0; 
ESTIMATE "Sex Diff at T1 if Ed=16" MW 1 MW*t1 0 MW*t1*t1 0  MW*ed12 4; 
 
ESTIMATE "Sex Diff at T6 if Ed=12" MW 1 MW*t1 5 MW*t1*t1 25 MW*ed12 0; 
ESTIMATE "Sex Diff at T6 if Ed=16" MW 1 MW*t1 5 MW*t1*t1 25 MW*ed12 4; 
 
 
ESTIMATE "Ed Slope at T1 for Men"  ed12 1 ed12*t1 0 MW*ed12 0;  
ESTIMATE "Ed Slope at T1 for Women" ed12 1 ed12*t1 0 MW*ed12 1; 
 
ESTIMATE "Ed Slope at T6 for Men"  ed12 1 ed12*t1 5 MW*ed12 0; 
ESTIMATE "Ed Slope at T6 for Women" ed12 1 ed12*t1 5 MW*ed12 1; 
 
 
* Example of asking for simple two-way interactions of a three-way interaction; 
 
ESTIMATE "Quadratic for Men" t1*t1 1 MW*t1*t1 0; 
ESTIMATE "Quadratic for Women"  t1*t1 1 MW*t1*t1 1;  
 
 
ESTIMATE "Sex by Time at T1" MW*t1 1 MW*t1*t1 0; 
ESTIMATE "Sex by Time at T6" MW*t1 1 MW*t1*t1 10; 
 
run; 
 
* Can check estimates by re-centering predictors; 
TITLE "Second Model: Time6=0 and Women=0 and Ed16=0 Instead"; 
PROC MIXED DATA=example COVTEST NOCLPRINT NOITPRINT METHOD=ML; 
 CLASS ID; 
 MODEL income = t6 t6*t6 WM ed16 WM*ed16  
      WM*t6 WM*t6*t6 ed16*t6 / SOLUTION DDFM=Satterthwaite;  
 RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;  
run; 

WHAT SAS GIVES YOU BY DEFAULT IF TIME1=0, MEN=0, AND ED12=0  
(same or different interpretation?): 

       Solution for Fixed Effects 
                         Standard 
Effect       Estimate       Error       
Intercept     49.9467     0.05817    (same as before) 
t1             4.0427     0.03066    at T1, income increases 4k/time for men with ed=12 
t1*t1         -0.2570    0.005886    linear rate of change in income decreases by .5/time in men 
MW            -9.9066     0.07503    if ed=12, women make 9.9k less than men at T1 
ed12           0.9451     0.02614    in men, income increases by .9/year of ed at T1 
MW*ed12       -0.3983     0.03269    (same as before) 
t1*MW          0.2074     0.03955    gender gap reduced by .2/time at T1 (for all ed) 
t1*t1*MW       0.1067    0.007592    quadratic change is .1 less negative in women 
t1*ed12        0.5320    0.002741    ed slope is more positive by .5/time (for both sexes) 
 

WHAT EXTRA CONDITIONAL PREDICTIONS AND EFFECTS WE ASKED FOR: 

                                     Estimates 
                                            Standard 
Label                           Estimate       Error       
Linear at T1 for Men   Ed=12      4.0427     0.03066     
Linear at T1 for Men   Ed=16      6.1709     0.03248     
Linear at T1 for Women Ed=12      4.2501     0.02498     
Linear at T1 for Women Ed=16      6.3782     0.02734     
Linear at T6 for Men   Ed=12      1.4727     0.03066     
Linear at T6 for Men   Ed=16      3.6008     0.03248     
Linear at T6 for Women Ed=12      2.7473     0.02498     
Linear at T6 for Women Ed=16      4.8754     0.02734     
Sex Diff at T1 if Ed=12          -9.9066     0.07503     
Sex Diff at T1 if Ed=16         -11.4998      0.1498     
Sex Diff at T6 if Ed=12          -6.2016     0.07503     
Sex Diff at T6 if Ed=16          -7.7948      0.1498     
Ed Slope at T1 for Men            0.9451     0.02614     
Ed Slope at T1 for Women          0.5468     0.02188     
Ed Slope at T6 for Men            3.6052     0.02614     
Ed Slope at T6 for Women          3.2069     0.02188     
Quadratic for Men                -0.2570    0.005886     
Quadratic for Women              -0.1503    0.004796     
Sex by Time at T1                 0.2074     0.03955     
Sex by Time at T6                 1.2747     0.03955 
 
WHAT YOU GET IF YOU RE-CENTER AT TIME6=0, WOMEN=0, AND ED16=0 INSTEAD: 

       Solution for Fixed Effects 
                         Standard 
Effect       Estimate       Error       
Intercept     70.3613      0.1006   (same as before)  
t6             4.8754     0.02734   at T6, income increases by 4.8k/time for women with ed=16 
t6*t6         -0.1503    0.004796   linear rate of change in income decreases by .3/time in women  
WM             7.7948      0.1498   if ed=16, men make 7.8k more than women at T6  
ed16           3.2069     0.02188   in women, income increases by 3.2k/year of ed at T6 
WM*ed16        0.3983     0.03269   (same as before)  
t6*WM         -1.2747     0.03955   gender gap reduced by 1.3k/time at T6 (for all ed)  
t6*t6*WM      -0.1067    0.007592   quadratic change is .1 more negative in men  
t6*ed16        0.5320    0.002741   ed slope is more positive by .5/time (for both sexes)  

************************** Model 3 ***************************************; 
TITLE "Third Model: Fixed Quadratic Time, Time1=0 and Men=0 and Ed12=0"; 
PROC MIXED DATA=example COVTEST NOCLPRINT NOITPRINT METHOD=ML; 
 CLASS ID; 
 MODEL income = t1 t1*t1 MW ed12 MW*ed12 MW*t1 MW*t1*t1 ed12*t1 
      MW*ed12*t1 / SOLUTION DDFM=Satterthwaite;  
 RANDOM INTERCEPT / SUBJECT=ID TYPE=UN; 
 
* Example of asking for slopes at particular occasions for particular people; 
* Here one of the "t1"=1, the other t1=actual time; 
ESTIMATE "Linear T1 M Ed=12"   t1 1 t1*t1 0  MW*t1 0 MW*t1*t1 0  ed12*t1 0 MW*ed12*t1 0; 
ESTIMATE "Linear T1 M Ed=16"   t1 1 t1*t1 0  MW*t1 0 MW*t1*t1 0  ed12*t1 4 MW*ed12*t1 0; 
ESTIMATE "Linear T1 W Ed=12" t1 1 t1*t1 0  MW*t1 1 MW*t1*t1 0  ed12*t1 0 MW*ed12*t1 0; 
ESTIMATE "Linear T1 W Ed=16" t1 1 t1*t1 0  MW*t1 1 MW*t1*t1 0  ed12*t1 4 MW*ed12*t1 4; 
 
ESTIMATE "Linear T6 M Ed=12" t1 1 t1*t1 10 MW*t1 0 MW*t1*t1 0  ed12*t1 0 MW*ed12*t1 0; 
ESTIMATE "Linear T6 M Ed=16" t1 1 t1*t1 10 MW*t1 0 MW*t1*t1 0  ed12*t1 4 MW*ed12*t1 0; 
ESTIMATE "Linear T6 W Ed=12" t1 1 t1*t1 10 MW*t1 1 MW*t1*t1 10 ed12*t1 0 MW*ed12*t1 0; 
ESTIMATE "Linear T6 W Ed=16" t1 1 t1*t1 10 MW*t1 1 MW*t1*t1 10 ed12*t1 4 MW*ed12*t1 4; 
 
* Example of asking for simple main effects of an interaction; 
* Here, "t1" is actual time always; 
ESTIMATE "Sex Diff at T1 if Ed=12"   MW 1 MW*ed12 0 MW*t1 0 MW*t1*t1 0  MW*ed12*t1 0; 
ESTIMATE "Sex Diff at T1 if Ed=16" MW 1 MW*ed12 4 MW*t1 0 MW*t1*t1 0  MW*ed12*t1 0; 
 
ESTIMATE "Sex Diff at T6 if Ed=12" MW 1 MW*ed12 0 MW*t1 5 MW*t1*t1 25 MW*ed12*t1 0; 
ESTIMATE "Sex Diff at T6 if Ed=16" MW 1 MW*ed12 4 MW*t1 5 MW*t1*t1 25 MW*ed12*t1 20; 
 
ESTIMATE "Ed Slope at T1 for Men"  ed12 1 MW*ed12 0 ed12*t1 0  MW*ed12*t1 0;  
ESTIMATE "Ed Slope at T1 for Women" ed12 1 MW*ed12 1 ed12*t1 0  MW*ed12*t1 0; 
 
ESTIMATE "Ed Slope at T6 for Men"  ed12 1 MW*ed12 0 ed12*t1 5  MW*ed12*t1 0; 
ESTIMATE "Ed Slope at T6 for Women" ed12 1 MW*ed12 1 ed12*t1 5  MW*ed12*t1 5;  
 
* Example of asking for simple two-way interactions of a three-way interaction; 
ESTIMATE "Quadratic for Men"       t1*t1 1 MW*t1*t1 0; 
ESTIMATE "Quadratic for Women"  t1*t1 1 MW*t1*t1 1;  
 
ESTIMATE "Sex by Ed at T1"    MW*ed12 1 MW*ed12*t1 0; 
ESTIMATE "Sex by Ed at T6"    MW*ed12 1 MW*ed12*t1 5; 
 
ESTIMATE "Ed by Time for Men"  ed12*t1 1 MW*ed12*t1 0; 
ESTIMATE "Ed by Time for Women"  ed12*t1 1 MW*ed12*t1 1; 
 
ESTIMATE "Sex by Time (at T1) for Ed=12" MW*t1  1 MW*ed12*t1 0 MW*t1*t1 0; 
ESTIMATE "Sex by Time (at T1) for Ed=16" MW*t1  1 MW*ed12*t1 4 MW*t1*t1 0; 
ESTIMATE "Sex by Time (at T6) for Ed=12" MW*t1  1 MW*ed12*t1 0 MW*t1*t1 10; 
ESTIMATE "Sex by Time (at T6) for Ed=16" MW*t1  1 MW*ed12*t1 4 MW*t1*t1 10; 
 
run; 
 
* Can check estimates by re-centering predictors; 
TITLE "Third Model: Time6=0 and Women=0 and Ed16=0 Instead"; 
PROC MIXED DATA=example COVTEST NOCLPRINT NOITPRINT METHOD=ML; 
 CLASS ID; 
 MODEL income = t6 t6*t6 WM ed16 WM*ed16 WM*t6 WM*t6*t6 ed16*t6  
      WM*ed16*t6 / SOLUTION DDFM=Satterthwaite;  
 RANDOM INTERCEPT / SUBJECT=ID TYPE=UN; 
run; 
 

WHAT SAS GIVES YOU BY DEFAULT IF TIME1=0, MEN=0, AND ED12=0  
(same or different interpretation?): 

       Solution for Fixed Effects 
                         Standard 
Effect       Estimate       Error     
Intercept      49.9393     0.05803   (same as before)  
t1              4.0457     0.03036   (same as before)   
t1*t1          -0.2570    0.005828   (same as before) 
MW             -9.9023     0.07484   (same as before) 
ed12            1.0275     0.02739   (same as before) 
MW*ed12        -0.5367     0.03549   gender gap grows by .5/year of ed at T1 OR ed slope is .5 less in W atT1  
t1*MW           0.2056     0.03916   gender gap reduced by .2/time at T1 for ed=12  
t1*t1*MW        0.1067    0.007517   (same as before)  
t1*ed12         0.4991    0.004268   ed slope is more positive by .5/time linearly for men  
t1*MW*ed12     0.05536    0.005529   reduction of gender gap of .2/time grows .05/year of ed OR increase in   
       ed slope over time is .05/time larger in women OR bigger gender gap per year of ed lessens by .05/time 
 

WHAT EXTRA CONDITIONAL PREDICTIONS AND EFFECTS WE ASKED FOR: 
                                     Estimates 
                                             Standard 
Label                            Estimate       Error       
Linear at T1 for Men Ed=12         4.0457     0.03036     
Linear at T1 for Men Ed=16         6.0419     0.03464     
Linear at T1 for Women Ed=12       4.2513     0.02473     
Linear at T1 for Women Ed=16       6.4689     0.02855     
Linear at T6 for Men Ed=12         1.4756     0.03036     
Linear at T6 for Men Ed=16         3.4718     0.03464     
Linear at T6 for Women Ed=12       2.7485     0.02473     
Linear at T6 for Women Ed=16       4.9662     0.02855     
Sex Diff at T1 if Ed=12           -9.9023     0.07484     
Sex Diff at T1 if Ed=16          -12.0490      0.1595     
Sex Diff at T6 if Ed=12           -6.2060     0.07484     
Sex Diff at T6 if Ed=16           -7.2456      0.1595     
Ed Slope at T1 for Men             1.0275     0.02739     
Ed Slope at T1 for Women           0.4908     0.02256     
Ed Slope at T6 for Men             3.5228     0.02739     
Ed Slope at T6 for Women           3.2629     0.02256     
Quadratic for Men                 -0.2570    0.005828     
Quadratic for Women               -0.1503    0.004748     
Sex by Ed at T1                   -0.5367     0.03549     
Sex by Ed at T6                   -0.2599     0.03549     
Ed by Time for Men                 0.4991    0.004268     
Ed by Time for Women               0.5544    0.003515     
Sex by Time (at T1) for Ed=12      0.2056     0.03916     
Sex by Time (at T1) for Ed=16      0.4270     0.04489     
Sex by Time (at T6) for Ed=12      1.2729     0.03916     
Sex by Time (at T6) for Ed=16      1.4943     0.04489     
 
WHAT YOU GET IF YOU RE-CENTER AT TIME6=0, WOMEN=0, AND ED16=0 INSTEAD: 
       Solution for Fixed Effects 
                         Standard 
Effect       Estimate       Error       
Intercept      70.5881      0.1030  (same as before)   
t6              4.9662     0.02855  (same as before)   
t6*t6          -0.1503    0.004748  (same as before)   
WM              7.2456      0.1595  (same as before)   
ed16            3.2629     0.02256  (same as before)   
WM*ed16         0.2599     0.03549 gender gap grows by .26/year of ed at T6 OR ed slope is .26 more in M atT6   
t6*WM          -1.4943     0.04489  gender gap reduced by 1.5/time at T6 for ed=16   
t6*t6*WM       -0.1067    0.007517  (same as before)  
t6*ed16         0.5544    0.003515  ed slope is more positive by .55/time linearly for women   
t6*WM*ed16    -0.05536    0.005529   


